

Programmation Fonctionnelle  
Cours 3  
Listes, filtrage par motif

October 4, 2012

Le type des listes

- ▶ Type prédéfini de OCaml
- ▶ En fait, il n'y a pas un seul type liste, mais il y a des listes d'entiers, de flottants, de booléens, etc.
- ▶ On dit que `list` est un type **polymorphe** : pour tout type  $t$ , il y a un type  $t\ \text{list}$ .
- ▶ On peut former des listes de n'importe quel type (fonctions ou listes incluses)

Exemples (lists1.ml)

```
[1; 2; 3];;  
[4.0; 7e2];;  
[42; 2.0; "toto"];; (* erreur de typage *)  
["ab"; "cd"];;  
[sin; cos];;  
[int_of_float; float_of_int];; (* erreur typage *)  
[1;2] = [3;4];;  
[1;2] = [1.0; 2.0];; (* erreur de typage *)  
[1, 2, 3];; (* pas confondre ; et , *)
```

Variable de type

- ▶ Les listes sont homogènes : tous les éléments doivent avoir le même type (qui est donc utilisé pour déterminer le type de liste)
- ▶ Quel est le type de la liste vide ?
- ▶ Il s'agit d'un type polymorphe '`list`'. '`list`' est une variable de type. Les variables de type commencent sur '`'`.

Exemples (lists2.ml)

```
[ ];; (* liste vide *)

[ ] < [ 1; 2 ];;

[ ] < [ "hello"; "goodbye" ];;
```

Exemples (lists3.ml)

```
(* comparaison de listes *)
[ 1; 2 ] = [ 1; 2 ];;

[ 1; 2 ] < [ 1; 2; 3 ];;

[ 1; 3 ] < [ 1; 2; 3 ];;

[ 1; 2; 3 ] < [ 1; 3 ];; (* ordre lexicographique *)
```

Exemples (lists4.ml)

```
(* Constructeur de liste *)
1 :: [ 2; 3 ];;

(* associe a droite *)
4 :: 5 :: [ 6; 7 ];;

(* erreur de typage *)
( 4 :: 5 ) :: [ 6; 7 ];;
```

Exemples (lists5.ml)

```
List.hd [ 1; 2; 3 ];;
List.tl [ 1; 2; 3 ];;
List.hd [];;
```

## Filtrage par motif

- ▶ angl. : **pattern matching**
- ▶ Très utile sur les type structurés, combine une distinction de cas avec un moyen facile de déconstruire une donnée
- ▶ Principe générale, pas seulement pour les listes.

## Exemples (lists6.ml)

```
match [4; 5] with
| [] -> print_string "vide"
| a::l -> print_string "pas_vide";;
```

```
let est_vide l = match l with
| [] -> true
| x -> false;;
```

```
let est_vide l = match l with
| [] -> true
| _ -> false;;
```

## Exemples (lists7.ml)

```
let tete l = match l with
| [] -> failwith "Liste_vide_dans_la_fonction_tete"
(* exception *)
| a::_ -> a;;
```

```
let queue l = match l with
| [] -> failwith "Liste_vide_dans_la_fonction_queue"
(* exception *)
| _::finliste -> finliste;;
```

*(\* quel est le type de ces deux fonctions ? \*)*

## Exemples (lists8.ml)

```
(* quel est l'erreur ? *)
let longueur l = match l with
| [] -> 0
| a::l' -> 1 + (longueur l');;
```

## Filtrage par motif

- ▶ Les motifs sont essayés dans l'ordre
- ▶ Si un motif s'applique, **toutes** les identificateurs dans le motif sont liés. Leur portée : l'expression à droite du motif.
- ▶ On peut dans un motif écrire `_` ou un identificateur préfixé par `_`, dans ce cas il n'y a pas de liaison.
- ▶ Les motifs doivent être linéaires (pas de répétition d'identificateur)
- ▶ Pas d'applications de fonctions dans les motifs

## Exemples (lists9.ml)

```
let rec print_sep l = match l with
| [] -> print_newline()
| [toto] -> print_int toto
| titi::toto::finliste ->
    print_int titi;
    print_char ' ';
    print_sep (toto::finliste);;

print_sep [1;2;3];;
```

## Exemples (lists10.ml)

```
let rec even_length l =
  match l with
  | [] -> true
  | [_] -> false
  | _::_::reste -> even_length reste
;;

even_length [1;2;3];;
even_length [1;2;3;4];;
```

## Exemples (lists11.ml)

```
(* erreur: motif non lineaire *)
let f l = match l with
| x::x::reste -> 1
| _ -> 0
;;
```

### Exemples (lists12.ml)

```

(* erreur: pas un motif *)

let f l = match l with
| 1+2::reste -> 3
| _ -> 5
;;

```

### Exemples (lists13.ml)

```

(* mauvais ordre de motifs *)

let f l = match l with
| [] -> print_string "vide"
| a::finliste -> print_int a
| [toto] -> print_int toto      (* jamais atteint *)
;;

```

### Exemples (lists14.ml)

```

(* Filtrage non exhaustif *)
let g l = match l with
| [] -> print_string "vide"
| [toto] -> print_char toto;;

g ['a'];;

g ['a'; 'b'];;

```

### Exemples (lists15.ml)

```

(* quel est l'erreur ? *)
let rec trouve a l = match l with
| [] -> false
| a::_ -> true
| b::_l -> trouve a l;;

trouve 1 [1;2;3];;
trouve 42 [1;2;3];;

```

### Exemples (lists16.ml)

```

(* la fonction corrige *)
let rec trouve a l = match l with
| [] -> false
| b::r -> if a=b then true else trouve a r
;;

trouve 1 [1;2;3];;
trouve 42 [1;2;3];;

```

### Exemples (lists17.ml)

```

(* Recherche de la sous-liste commençant par a *)
let rec trouve_sous_liste a l = match l with
| [] -> []
| p::ll -> if p = a then l else trouve_sous_liste a ll

(* Avec une fonction auxiliaire pour éviter de tout calculer *)
let trouve_sous_liste a l =
  let rec aux l = match l with
  | [] -> []
  | p::ll -> if p = a then l else aux ll
  in aux l
;;

trouve_sous_liste 3 [5;6;7;8;9];;

```

### Autres fonctions utiles

- `List.length` donne la longueur d'une liste
- Opérateur `@` : concaténation de deux listes.
- Attention, `@` n'est pas un constructeur mais une fonction définie par récurrence.
- Plus de fonctions : voir la bibliothèque `List`

### Exemples (append.ml)

```

let rec append l1 l2 = match l1 with
| [] -> l2
| h1::r1 -> h1 :: (append r1 l2)
;;

append [1;2;3;4] [5;6;7;8];;

```

## Plus sur le filtrage par motif

- ▶ S'applique à n'importe quel type (sauf fonctions et objets)
- ▶ Motifs avec des alternatives
- ▶ Motifs avec des conditions

## Exemples (motif1.ml)

```

let rec fib n = match n with
| 0 -> 0
| 1 -> 1
| n -> fib (n-1) + fib (n-2)
;;

fib 10;;

```

## Exemples (motif2.ml)

```

(* motif avec des alternatives *)
let rec fib n = match n with
| 0 | 1 -> n
| n -> fib (n-1) + fib (n-2)
;;

fib 10;;

```

## Exemples (motif3.ml)

```

(* la fonction trouve corrige *)
let rec trouve a l = match l with
| [] -> false
| b::r when b=a -> true
| _::r -> trouve a r
;;

trouve 1 [1;2;3];;
trouve 42 [1;2;3];;

```