
TP n
�
5 - Correction

Expressions symboliques

L'objectif de ce TP est de manipuler des expressions mathématiques à plusieurs symboles, pour
les évaluer, les simpli�er et les dériver.

1 Dé�nition

Une expression à plusieurs symboles, ou expression symbolique, est un objet mathématique qui
peut être :
� soit un nombre �ottant
� soit l'une des formes suivantes : C1 + C2, C1 � C2, C1 � C2, cosC1, sinC2, où C1 et C2 sont
deux expressions symboliques

� soit un symbole (x, y, ab, etc.)
Par exemple :

((2: � 1:) � (1: � x)) � ((19: � cos(0:)) + sin(theta))

est une expression qui contient deux symboles, x et theta.

Exercice 1 On donne les types suivants pour représenter les opérateurs binaires et unaires :

type binop = Plus | Moins | Fois

type unop = Cos | Sin

Dé�nir le type expr pour représenter les expressions symboliques, sous la forme d'un type somme
à quatre constructeurs : Nombre, Binop, Unop et Symbole (un symbole étant représenté par une
chaîne de caractères), de sorte que l'expression e0 ci-dessus sera représentée en OCaml par :

let e0 = Binop (Fois,

Binop (Moins,

Binop (Fois, Nombre 2., Nombre 1.),

Binop (Fois, Nombre 1., Symbole "x")),

Binop (Plus,

Binop (Moins, Nombre 19., Unop (Cos, Nombre 0.)),

Unop (Sin, Symbole "theta"))) ;;

Correction :

type expr =

| Nombre of float

| Binop of (binop * expr * expr)

| Unop of (unop * expr)

| Symbole of string

Exercice 2 Écrire une fonction :

string_of_expr : expr -> string

qui, étant donnée une expression, renvoie une chaîne de caractères représentant l'expression
entièrement parenthésée sous forme lisible par un être humain. Par exemple, pour e0, cette
fonction renverra la chaîne :

"(((2.*1.)-(1.*x))*((19.-cos(0.))+sin(theta)))"

On pourra écrire au préalable des fonctions :
� string_of_binop : binop -> string

� string_of_unop : unop -> string

qui étant donné un opérateur, renvoient sa représentation lisible en chaîne de caractères.

Correction :

let string_of_binop = function

| Plus -> "+"

| Moins -> "-"

| Fois -> "*"

let string_of_unop = function

| Cos -> "cos"

| Sin -> "sin"

let rec string_of_expr = function

| Nombre n -> string_of_float n

| Binop (b, e1, e2) -> "(" ^ string_of_expr e1 ^ string_of_binop b ^ string_of_expr e2 ^ ")"

| Unop (u, e) -> string_of_unop u ^ "(" ^ string_of_expr e ^ ")"

| Symbole x -> x

2 Évaluation

Exercice 3 Écrire une fonction :

eval_expr : (string * float) list -> expr -> float

telle que, si l est un environnement associant (sous la forme d'une liste d'association) chaque
symbole à une valeur �ottante, et si e est une expression, alors eval_expr l e calcule et renvoie
la valeur de l'expression e sous l'environnement l. Par exemple, pour évaluer l'expression e0
ci-dessus en prenant 3: pour x et 0: pour theta, on écrira :

eval_expr [("x", 3.); ("theta", 0.)] e0

ce qui renverra -18. (car cos 0 = 1 et sin 0 = 0).
Indication : on pourra utiliser la fonction List.assoc : 'a -> ('a * 'b) list -> 'b.
On pourra au préalable écrire des fonctions :
� valeur_binop : binop -> float -> float -> float

� valeur_unop : unop -> float -> float

qui e�ectuent les opérations binaires et unaires en supposant que leurs arguments sont des valeurs
�ottantes déjà calculées.

Correction :

let valeur_binop = function

| Plus -> ( +. )

| Moins -> ( -. )

| Fois -> ( *. )

let valeur_unop = function

| Cos -> cos

| Sin -> sin

let rec eval_expr l = function

| Nombre n -> n

| Binop (b, e1, e2) -> valeur_binop b (eval_expr l e1) (eval_expr l e2)

| Unop (u, e) -> valeur_unop u (eval_expr l e)

| Symbole x -> List.assoc x l

3 Simpli�cation

Exercice 4 Écrire une fonction :

eval_sous_expr : expr -> expr

telle que, si e est une expression, alors eval_sous_expr e renvoie l'expression obtenue en rem-
plaçant toute sous-expression de e ne contenant pas de symbole, par sa valeur. Cette passe est
appelée évaluation des sous-expressions constantes.
Par exemple, dans l'expression e0 ci-dessus, les sous-expressions 2:�1: et 19:�cos 0: ne contiennent
pas de symbole, elles pourront donc être remplacées par leurs valeurs, et alors eval_expr_partielle
e0 renverra l'expression : (2: � (1: � x)) � (18: + sin(theta))
Faire des essais pour remarquer que, si e est une expression quelconque, alors, pour tout environ-
nement l, eval_expr l e et eval_expr l (eval_sous_expr e) renvoient les mêmes valeurs.
On dit que eval_sous_expr est correcte vis-à-vis de l'évaluation des expressions.
Remarquer aussi que pour toute expression e :

eval_sous_expr(eval_sous_expr e) = eval_sous_expr e

On dit que eval_sous_expr est idempotente.

Correction :

let rec eval_sous_expr = function

| Binop (b, e1, e2) -> begin match (eval_sous_expr e1, eval_sous_expr e2) with

| Nombre n1, Nombre n2 -> Nombre (valeur_binop b n1 n2)

| (e'1, e'2) -> Binop (b, e'1, e'2)

end

| Unop (u, e) -> begin match eval_sous_expr e with

| Nombre n -> Nombre (valeur_unop u n)

| e' -> Unop (u, e')

end

| e -> e

Exercice 5 Écrire une fonction :

eval_neutres : expr -> expr

telle que, si e est une expression, alors eval_neutres e renvoie l'expression obtenue en rempla-
çant toute sous-expression de e :
� de la forme 0: � e′ ou e′ � 0: ou e′ � e′ par 0:
� de la forme 1 � e′, e′ � 1 ou e′ � 0 ou e′ + 0: ou 0: + e′ par e′

Par exemple, dans l'expression e0 ci-dessus, la sous-expression 2 � 1 (resp. 1 � x) pourra être
remplacée par 2 (resp. x), et alors eval_neutres e0 renverra l'expression :

(2: � x) � ((19: � cos(0:)) + sin(theta))

(l'évaluation des sous-expressions constantes n'a pas lieu)
Remarquer que eval_neutres est correcte vis-à-vis de l'évaluation des expressions, et qu'elle est
idempotente.

Correction :

let rec eval_neutres = function

| Binop (b, e1, e2) ->

begin match b, eval_neutres e1, eval_neutres e2) with

| Moins, e'1 , e'2 when e'1 = e'2 -> Nombre 0.

| Fois , _ , Nombre 0. -> Nombre 0.

| Fois , Nombre 0., _ -> Nombre 0.

| Fois , e' , Nombre 1. -> e'

| Fois , Nombre 1., e' -> e'

| Moins, e' , Nombre 0. -> e'

| Plus , e' , Nombre 0. -> e'

| Plus , Nombre 0., e' -> e'

| _ , e'1 , e'2 -> Binop (b, e'1, e'2)

end

| Unop (u, e') -> Unop (u, eval_neutres e')

| e -> e

Exercice 6 Écrire une fonction :

point_fixe : ('a -> 'a) -> 'a -> 'a

telle que, si f est une fonction et si a est une valeur applicable à f , alors point_fixe f a

applique f sur a et recommence sur le résultat jusqu'à ce qu'il ne change plus.
On dit que point_fixe f a est un point �xe de f (en e�et, si x = point_fixe f a, alors
f x = x). C'est le point �xe de f obtenu en partant de a.
Utiliser cette fonction pour écrire une fonction :

simplifier : expr -> expr

qui, étant donnée une expression, lui applique les deux fonctions eval_neutres et eval_sous_expr
et continue jusqu'à ce que l'expression ne change plus.
Pourquoi cette fonction termine-t-elle ?

Correction :

let rec point_fixe f a =

let a' = f a in

if a' = a

then a

else point_fixe f a'

let simplifier = point_fixe (fun e -> let e' = eval_neutres e in eval_sous_expr e')

Cette fonction termine car chacune des deux fonctions eval_neutres et eval_sous_expr, soit laisse
l'expression inchangée, soit fait décroître strictement sa taille (nombre de constructeurs, par exemple).

4 Dérivation

Exercice 7 Soit s un symbole. L'expression dérivée d'une expression e par rapport au symbole
s, notée ∂e

∂s ou @se, est dé�nie par induction sur la structure de e :
� si n est un nombre (exact ou approché), alors sa dérivée est le nombre 0:
� si e1 et e2 sont deux expressions, alors @s(e1 + e2) = @se1 + @se2 et de même pour �
� si e1 et e2 sont deux expressions, alors @s(e1 � e2) = (@se1 � e2) + (e1 � @se2)
� si e est une expression, alors @s(cos(e)) = 0: � (@se � sin(e)) et @s(sin(e)) = @se � cos(e))
� la dérivée du symbole s par rapport à lui-même est le nombre 1:
� la dérivée d'un symbole t di�érent de s est le nombre 0:
Écrire une fonction :

deriv_expr : string -> expr -> expr

qui, étant donnés un symbole s et une expression e, calcule et renvoie l'expression dérivée de e
par rapport à s.

Correction :

let rec deriv_expr s = function

| Nombre _ -> Nombre 0.

| Binop (b, e1, e2) ->

let e'1 = deriv_expr s e1

and e'2 = deriv_expr s e2

in

begin match b with

| Fois -> Binop (Plus, Binop (Fois, e'1, e2), Binop (Fois, e1, e'2))

| _ -> Binop (b, e'1, e'2)

end

| Unop (u, e) ->

let e' = deriv_expr s e

in

begin match u with

| Cos -> Binop (Moins, Nombre 0., Binop (Fois, e', Unop (Sin, e)))

| Sin -> Binop (Fois, e', Unop (Cos, e))

end

| Symbole t -> Nombre (if t = s then 1. else 0.)

